
ПИИС – Упражнения – Модул 9

**Двумерни масиви. Задачи с матрици.
Суми, произведения, минимум, максимум,
работа с елементи под / над главен /
второстепенен диагонал.**



доц. д-р арх. Стоянка Иванова
р-л катедра „Автоматизация на инженерния труд“
Строителен факултет, УАСГ

Email: siva_fce@uacg.bg

Условие: Дадена е матрица $Z(M,N)$ с максимум 8 реда и 8 стълба, съставена от елементи от тип **Double**. Записана е в **Sheet1**. Да се намерят:

- а) сумата на всички елементи;
- б) произведението на всички ненулеви елементи в матрицата;
- в) максимума на матрицата и местоположението му (ред и стълб);
- г) минимума на матрицата и местоположението му (ред и стълб);
- д) средното аритметично на положителните елементи над главния диагонал, ако матрицата е квадратна ($M=N$);
- е) максимума под второстепенния диагонал, ако матрицата е квадратна;
- ж) сумите по редове на матрицата;
- з) минимумите по стълбове на матрицата;
- и) да се копират всички отрицателни елементи в едномерния масив **D** и да се подредят във възходящ ред. Да се изведат трите най-малки от тях.

	1	2	3	4
1	4	1	2	-4
2	-4	5	-10	5
3	6	4	-9	-2
4	7	6	-3	9

Option Explicit

' Процедура за работа с матрица

```
Sub Matrix()
```

```
    Dim Z() As Double
```

```
    Dim M As Integer, N As Integer
```

```
    Dim i As Integer, j As Integer
```

' зареждане на данни от Excel -> Sheet1

```
    Dim W As Worksheet
```

```
    Set W = Application.Worksheets("Sheet1") ' така може да се избират различни страници
```

```
    W.Activate
```

```
    Do
```

```
        M=InputBox("M=")
```

```
    Loop Until M>1 and M<=8
```

```
    Do
```

```
        N=InputBox("N=")
```

```
    Loop Until N>1 and N<=8
```

```
    ReDim Z(1 To M, 1 To N)
```

```
    For i = 1 To M
```

```
        For j = 1 To N
```

```
            Z(i, j) = W.Cells(i, j)
```

```
        Next
```

```
    Next
```

	1	2	3	4
1	4	1	2	-4
2	-4	5	-10	5
3	6	4	-9	-2
4	7	6	-3	9

- Масивът Z() е динамичен, броят редове и броят стълбове се определят след въвеждане на M и N, с ReDim.
- Стойностите на масива Z() се вземат от страница Sheet1, трябва да започват от клетка A1.

' а) сума на всички елементи в матрицата

```
Dim S As Double
S = 0
For i = 1 To M
    For j = 1 To N
        S = S + Z(i, j)
    Next
Next
MsgBox ("S=" & S)
```

	1	2	3	4
1	4	1	2	-4
2	-4	5	-10	5
3	6	4	-9	-2
4	7	6	-3	9

- Променливата S е от същия тип, като елементите от масива Z .
- При изчисляване на сума, S се занулява.
- Обхождане по редове – затова първият цикъл е за i : `For i = 1 To M`
- В рамките на всеки ред се обхождат елементите отляво надясно, затова вторият цикъл е за j : `For j = 1 To N`
- Редът `S = S + Z(i,j)` означава, че в променливата S се натрупват стойностите на $Z(i,j)$.
- Накрая след циклите на екрана с помощта на функцията `MsgBox` се извежда стойността на сумата S .

' б) произведение на всички ненулеви елементи в матрицата

```
Dim P As Double
P = 1
For i = 1 To M
    For j = 1 To N
        If Z(i, j) <> 0 Then P = P * Z(i, j)
    Next
Next
MsgBox ("P=" & P)
```

	1	2	3	4
1	4	1	2	-4
2	-4	5	-10	5
3	6	4	-9	-2
4	7	6	-3	9

- Променливата P е от същия тип, като елементите от масива Z.
- При изчисляване на произведение, P получава стойност 1.
- Обхождане по редове – затова първият цикъл е за i: For i = 1 To M
- В рамките на всеки ред се обхождат елементите отляво надясно, затова вторият цикъл е за j: For j = 1 To N
- Редът If Z(i, j) <> 0 Then P = P * Z(i, j) означава, че в променливата P се умножават стойностите на ненулевите Z(i,j).
- Накрая след циклите на екрана с помощта на функцията MsgBox се извежда стойността на произведението P.

' в) максимум на матрицата и местоположението му (ред и стълб)

```
Dim Max As Double
Dim imax As Integer, jmax As Integer
Max = Z(1, 1)
imax = 1: jmax = 1
For i = 1 To M
    For j = 1 To N
        If Z(i, j) > Max Then Max = Z(i, j): imax = i: jmax = j
    Next
Next
MsgBox ("Max=Z(" & imax & ", " & jmax & ")=" & Max)
```

	1	2	3	4
1	4	1	2	-4
2	-4	5	-10	5
3	6	4	-9	-2
4	7	6	-3	9

- Променливата Max е от същия тип, като елементите от масива Z.
- Променливите imax и jmax (за номер на ред и номер на стълб на максимума) са цели (тип Integer).
- Променливите Max = Z(1, 1): imax = 1: jmax = 1 означават, че търсенето почва от горния ляв ъгъл на матрицата.
- Обхождане по редове – затова първият цикъл е за i: For i = 1 To M
- В рамките на всеки ред се обхождат елементите отляво надясно, затова вторият цикъл е за j: For j = 1 To N
- Когато намерим елемент $Z(i, j) > \text{Max}$, запомняме: $\text{Max} = Z(i, j)$: $\text{imax} = i$: $\text{jmax} = j$
- Накрая след циклите на екрана с помощта на функцията MsgBox се извежда стойността на максимума Max и местоположението му чрез imax и jmax.

' г) минимум на матрицата и местоположението му (ред и стълб)

```
Dim Min As Double
Dim imin As Integer, jmin As Integer
Min = Z(1, 1)
imin = 1: jmin = 1
For i = 1 To M
    For j = 1 To N
        If Z(i, j) < Min Then Min = Z(i, j): imin = i: jmin = j
    Next
Next
MsgBox ("Min=Z(" & imin & ", " & jmin & ")=" & Min)
```

	1	2	3	4
1	4	1	2	-4
2	-4	5	-10	5
3	6	4	-9	-2
4	7	6	-3	9

- Променливата *Min* е от същия тип, като елементите от масива *Z*.
- Променливите *imin* и *jmin* (за номер на ред и номер на стълб на минимума) са цели (тип *Integer*).
- Променливите *Min = Z(1, 1): imin = 1: jmin = 1* означават, че търсенето почва от горния ляв ъгъл на матрицата.
- Обхождане по редове – затова първият цикъл е за *i*: `For i = 1 To M`
- В рамките на всеки ред се обхождат елементите отляво надясно, затова вторият цикъл е за *j*: `For j = 1 To N`
- Когато намерим елемент $Z(i, j) < Min$, запомняме: `Min = Z(i, j): imin = i: jmin = j`
- Накрая след циклите на екрана с помощта на функцията `MsgBox` се извежда стойността на минимума *Min* и местоположението му чрез *imin* и *jmin*.

' д) средно аритметично на положителните елементи над главния диагонал, ако матрицата е квадратна

- Матрицата е квадратна, при $M = N$
- Главен диагонал:
 - По главния диагонал, ако $i = j$
 - Над главния диагонал, ако $i < j$
 - Под главния диагонал, ако $i > j$

	1	2	3	4
1	Z_{11}	Z_{12}	Z_{13}	Z_{14}
2	Z_{21}	Z_{22}	Z_{23}	Z_{24}
3	Z_{31}	Z_{32}	Z_{33}	Z_{34}
4	Z_{41}	Z_{42}	Z_{43}	Z_{44}

- Второстепенен диагонал:
 - По второстепенния диагонал, ако $i + j = M + 1$
 - Над второстепенния диагонал, ако $i + j < M + 1$
 - Под второстепенния диагонал, ако $i + j > M + 1$

	1	2	3	4
1	Z_{11}	Z_{12}	Z_{13}	Z_{14}
2	Z_{21}	Z_{22}	Z_{23}	Z_{24}
3	Z_{31}	Z_{32}	Z_{33}	Z_{34}
4	Z_{41}	Z_{42}	Z_{43}	Z_{44}

' д) средно аритметично на положителните елементи над главния диагонал, ако матрицата е квадратна

```

If M = N Then
  Dim Sp As Double, Ap As Double
  Dim Np As Integer
  Sp = 0: Np = 0
  For i = 1 To M
    For j = 1 To N
      If i < j Then
        If Z(i, j) > 0 Then Sp = Sp + Z(i, j): Np = Np + 1
      End If
    Next
  Next
  If Np > 0 Then
    Ap = Sp / Np
    MsgBox ("Ap=" & Ap)
  Else
    MsgBox ("Няма положителни елементи над главния диагонал.")
  End If
End If
    
```

	1	2	3	4
1	Z ₁₁	1	2	-4
2	-4	Z ₂₂	-10	5
3	6	4	Z ₃₃	-2
4	7	6	-3	Z ₄₄

- Променливата Sp е за сума, Np е за брой на положителните елементи над главния диагонал.
- Обхождане по редове – затова първият цикъл е за i: For i = 1 To M
- Над главния диагонал (при i < j) проверяваме за Z(i, j) > 0 и натрупваме в променливата Sp, броим в Np.
- Накрая при Np > 0 изчисляваме средно аритметично Ap, иначе извеждаме съобщение, че няма положителни елементи над главния диагонал.

' е) максимум под второстепенния диагонал, ако матрицата е квадратна

```

If M = N Then
  Dim Max2 As Double
  Max2 = Z(M, N)
  For i = 1 To M
    For j = 1 To N
      If i + j > N + 1 Then
        If Z(i, j) > Max2 Then Max2 = Z(i, j)
      End If
    Next
  Next
  MsgBox ("Max2=" & Max2)
End If
    
```

	1	2	3	4
1	4	1	2	Z ₁₄
2	-4	5	Z ₂₃	5
3	6	Z ₃₂	-9	-2
4	Z ₄₁	6	-3	9

- Променливата Max2 е от тип Double, като елементите на масива Z.
- Започваме търсенето от Max2=Z(M,N), понеже този елемент е позициониран под второстепенния диагонал. Може да се почне и от друг елемент, стига да е под второстепенния диагонал.
- Обхождане по редове – затова първият цикъл е за i: For i = 1 To M
- Под второстепенния диагонал (при $i + j > N + 1$) проверяваме за елемент $Z(i, j) > \text{Max2}$ и го запомняме в Max2.
- Накрая извеждаме Max2.

' ж) суми по редове на матрицата

```
Dim SR() As Double
ReDim SR(1 To M)
For i = 1 To M
    SR(i) = 0
    For j = 1 To N
        SR(i) = SR(i) + Z(i, j)
    Next
    MsgBox ("SR(" & i & ")=" & SR(i))
Next
```

	1	2	3	4
1	4	1	2	-4
2	-4	5	-10	5
3	6	4	-9	-2
4	7	6	-3	9

- Масивът SR() е за **суми по редове**. Предекларираме го, за да има M елемента: ReDim SR(1 To M)
- Обхождане по редове – затова първият цикъл е за i: For i = 1 To M
- В началото на всеки ред i нулираме SR(i) = 0
- За всеки елемент в ред i сумираме SR(i) = SR(i) + Z(i, j)
- В края на всеки ред i извеждаме SR(i), преди да завърши вторият цикъл!
- Така по тази задача извеждаме 4 суми за 4 реда.

'з) минимуми по стълбове на матрицата

```
Dim MinK() As Double
ReDim MinK(1 To N)
For j = 1 To N
    MinK(j) = Z(1, j)
    For i = 2 To M
        If Z(i, j) < MinK(j) Then MinK(j) = Z(i, j)
    Next
    MsgBox ("MinK(" & j & ")=" & MinK(j))
Next
```

	1	2	3	4
1	4	1	2	-4
2	-4	5	-10	5
3	6	4	-9	-2
4	7	6	-3	9

- Масивът `MinK()` е за минимуми по стълбове. Предекларираме го, за да има `N` елемента: `ReDim MinK(1 To N)`
- Обхождане **по стълбове** – затова първият цикъл е за `j`: `For j = 1 To N`
- В началото на всеки стълб `j` стартираме с `MinK(j) = Z(1,j)`
- За всеки следващ елемент в стълб `j` проверяваме дали `Z(i, j) < MinK(j)` и ако да, го запомняме като `MinK(j)`.
- В края на всеки стълб `j` извеждаме `MinK(j)`, преди да завърши вторият цикъл.
- Така по тази задача извеждаме 4 минимума за 4 стълба.

' и) да се изведат 3те най-малки отрицателни елементи в матрицата, като се използва помощен едномерен масив D

```
Dim D( ) As Double
ReDim D(1 To M * N)
Dim L As Integer
L = 0
For i = 1 To M
    For j = 1 To N
        If Z(i, j) < 0 Then L = L+1: D(L) = Z(i, j)
    Next
Next
```

	1	2	3	4
1	4	1	2	-4
2	-4	5	-10	5
3	6	4	-9	-2
4	7	6	-3	9

- Масивът D() е помощен, в него копираме всички отрицателни елементи на матрицата и после подреждаме (сортираме) елементите му.
- Предекларираме D, да има M*N елемента ReDim D(1 To M * N), в случай, че всички елементи са отрицателни.
- В променливата L броим намерените отрицателни елементи.
- Обхождане по редове – затова първият цикъл е за i: For i = 1 To M
- За всеки отрицателен елемент в матрицата, увеличаваме L с 1 и D(L) = Z(i, j)

- В резултат получаваме едномерен масив D с $L=6$ елемента, подредени, както са срещнати при обхождането по редове:

1	2	3	4	5	6
-4	-4	-10	-9	-2	-3

' сортиране на масива D във възходящ ред по метода на мехурчето, едномерен масив се сортира чрез 2 цикъла!!!

```
Dim C As Double ' спомагателна променлива
```

```
For i = 1 To L - 1
```

```
  For j = 1 To L - i
```

```
    If D(j) > D(j + 1) Then C = D(j + 1): D(j + 1) = D(j): D(j) = C ' ако е нужно, разменяме стойностите на D(j) и D(j+1)
```

```
  Next
```

```
Next
```

- След подреждането във възходящ ред на масива D , получаваме следните стойности, от които извеждаме първите три:

1	2	3	4	5	6
-10	-9	-4	-4	-3	-2

' извеждане на трите най-малки елемента на масива D

```
If L >= 3 Then
```

```
  For i = 1 To 3
```

```
    MsgBox ("D(" & i & ")=" & D(i))
```

```
  Next
```

```
End If
```

```
End Sub
```

Внимание! Вижте илюстриращото видео тук: <https://youtu.be/rweRoWMAH4s>

Видеото илюстрира реализацията на условията на задачата на езика VBA.

Изгледайте видеото на части (с паузиране и анализ), към всяка от съответните части (подточки а, б, в...) от заданието, за да вникнете в начина, по който се обхожда матрицата, как се променят управляващите променливи i и j и спомагателните и резултантните променливи и масиви.

Задачата с матрица е много важна, защото:

- Курсовата задача, въз основа на която заверяваме, е с матрица;
- Задачата на изпит е с матрица;
- Задачата на контролно (при редовни студенти) е с матрица.

Допълнителни подусловия за упражнение на придобитите знания:

и1) към поусловие и) добавете извеждане на масива D , преди сортирането му.

й) изчислете и изведете произведения по стълбове (в нов масив PK).

к) намерете и изведете максимуми по редове (в нов масив $MaxR$).

Успех!

доц. д-р арх. Стоянка Иванова
р-л катедра „Автоматизация на инженерния труд“
Строителен факултет, УАСГ



Email siva_fce@uacg.bg

Страница: <https://uacg.bg/?p=178&l=1&id=215&f=2&dp=27>

Учебник по VBA: https://uacg.bg/filebank/att_11582.pdf

Кабинет 6, ет. 3 (над главното фоайе) в топлата връзка между корпус А и Б на УАСГ.